
⭐⭐⭐ Challenge ⏱️ 30 minutes

Deluxe parking sensor

The deluxe version of the parking sensor! It beeps faster when you get closer, and the OLED

screen shows the distance in centimeters. With a little bar that gets longer when you get

closer.

🧰 What do you need?

1× Arduino UNO 1× Distance sensor HC-

SR04

1× Active buzzer

1× OLED screen 0.91"

🔌
USB cable + Jumper

wires

🗺️ Wiring diagram

🛠️ This is how you build it

1. Connect the HC-SR04: Trig to pin 9, Echo to pin 10, Vcc to 5V and Gnd to

GND.

2. Connect the buzzer: S to pin 8, + to 5V and − to GND.

3. Connect the OLED screen: SDA to A4, SCL to A5, VCC to 5V and GND to

GND.

4. Three parts need 5V and GND. Use the + and − strips on your breadboard.

💻 The code

🧠 What does the code do?

measureDistance()  measures the distance in centimeters with the HC-

SR04.

showDistance()  puts the number big on the screen and draws a little bar

with fillRect() .

The buzzer beeps faster as the distance gets smaller.

🎯 Try this…

Show "STOP!" on the screen when you are closer than 5 cm

Show the distance in meters when something is more than 1 meter away

Add the LED ring that changes from green to red

🚀 Build this project with Codey: open codey.online/starter-kits/uno/projects/parking-sensor-deluxe and click MAKE THIS. Then the code and the wiring

diagram are ready right away.

Arduino UNO

Distance sensor HC-SR04 Active buzzer

OLED screen 0.91"

// Deluxe parking sensor

// Just like the normal parking sensor, but now you also see the distance on 

the OLED screen.

#include <Wire.h>

#include <Adafruit_GFX.h>

#include <Adafruit_SSD1306.h>

const int TRIG_PIN = 9;     // Trig of the HC-SR04 on pin 9

const int ECHO_PIN = 10;    // Echo of the HC-SR04 on pin 10

const int BUZZER_PIN = 8;   // S of the buzzer on pin 8

Adafruit_SSD1306 display(128, 32, &Wire, -1);  // OLED: SDA = A4, SCL = A5

// Measure the distance in centimeters

long measureDistance() {

  digitalWrite(TRIG_PIN, LOW);

  delayMicroseconds(2);

  digitalWrite(TRIG_PIN, HIGH);

  delayMicroseconds(10);

  digitalWrite(TRIG_PIN, LOW);

  long time = pulseIn(ECHO_PIN, HIGH, 30000);

  if (time == 0) return 999;   // No echo: nothing nearby

  return time / 58;

}

// Put the distance big on the screen, with a little bar below it

void showDistance(long distance) {

  display.clearDisplay();

  display.setTextSize(2);

  display.setCursor(0, 0);

  if (distance >= 999) {

    display.print(F("---"));

  } else {

    display.print(distance);

    display.print(F(" cm"));

  }

  // Little bar: the closer, the longer

  int length = map(constrain(distance, 0, 100), 0, 100, 128, 0);

  display.fillRect(0, 24, length, 8, SSD1306_WHITE);

  display.display();

}

void setup() {

  pinMode(TRIG_PIN, OUTPUT);

  pinMode(ECHO_PIN, INPUT);

  pinMode(BUZZER_PIN, OUTPUT);

  display.begin(SSD1306_SWITCHCAPVCC, 0x3C);

  display.setTextColor(SSD1306_WHITE);

}

void loop() {

  long distance = measureDistance();

  showDistance(distance);

  if (distance < 5) {

    digitalWrite(BUZZER_PIN, HIGH);   // Very close: one long tone

    delay(100);

  } else if (distance < 50) {

    digitalWrite(BUZZER_PIN, HIGH);   // Short beep

    delay(50);

    digitalWrite(BUZZER_PIN, LOW);

    delay(distance * 10);              // Closer = beeps faster

  } else {

    digitalWrite(BUZZER_PIN, LOW);

    delay(100);

  }

}

http://app/starter-kits/uno/tutorial/kit/hc-sr04
http://app/starter-kits/uno/tutorial/kit/active-buzzer-module
http://app/starter-kits/uno/tutorial/kit/oled-display-091-inch

